// ECB/SIS Public License, version 1.0, document reference SIS/2001/116
//
// Copyright (C) 2008 European Central Bank. All rights reserved.
//
// Redistribution and use in source and binary forms,
// with or without modification, are permitted
// provided that the following conditions are met:
//
// Redistributions of source code must retain the above copyright notice,
// this list of conditions and the following disclaimer.
// Redistributions in binary form must reproduce the above copyright notice,
// this list of conditions and the following disclaimer in the documentation
// and/or other materials provided with the distribution.
// Neither the name of the European Central Bank
// nor the names of its contributors may be used to endorse or promote products
// derived from this software without specific prior written permission.
//
// THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS
// "AS IS" AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED
// TO, THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A
// PARTICULAR PURPOSE ARE DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT
// HOLDERS OR CONTRIBUTORS BE LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL,
// SPECIAL, EXEMPLARY, OR CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT
// LIMITED TO, PROCUREMENT OF SUBSTITUTE GOODS OR SERVICES; LOSS OF USE,
// DATA, OR PROFITS; OR BUSINESS INTERRUPTION) HOWEVER CAUSED AND ON ANY
// THEORY OF LIABILITY, WHETHER IN CONTRACT, STRICT LIABILITY, OR TORT
// (INCLUDING NEGLIGENCE OR OTHERWISE) ARISING IN ANY WAY OUT OF THE USE OF
// THIS SOFTWARE, EVEN IF ADVISED OF THE POSSIBILITY OF SUCH DAMAGE.
package eu.ecb.core.view.util
{
	import mx.core.UIComponent;
	
	/**
	 * Factory in charge of creating PopUpMediators. The factory will maintain
	 * a list of such mediators (one per view which requests it).
	 * 
	 * The class is implemented as a singleton.
	 * 
	 * @author Xavier Sosnovsky
	 */ 
	public class PopUpMediatorFactory
	{
		/*==============================Fields================================*/
		
		 /** The single instance of the class */
        private static var _instance:PopUpMediatorFactory;
		
		/**
		 * The mediators already in use by the application. The mediator key
		 * is the uid of the view. This guarantees that the needed controllers 
		 * are only created once.
		 */
		 private var _mediators:Object; 
		 
		/*===========================Constructor==============================*/
		
		public function PopUpMediatorFactory(enforcer:SingletonEnforcer) {
			_mediators = new Object();
		}			
		
		/*==========================Public methods============================*/
		
		/**
		 * Get the factory for managing popup mediators 
		 * 
		 * @return The requested factory 
		 */
		static public function getInstance():PopUpMediatorFactory 
		{
			if (null == PopUpMediatorFactory._instance) {
				PopUpMediatorFactory._instance = 
					new PopUpMediatorFactory(new SingletonEnforcer());
			}
			return PopUpMediatorFactory._instance;
		}
		
		/**
		 * Get a mediator for managing popup windows 
		 * 
		 * @param view The view that will display the popup window
		 * 
		 * @return The mediator 
		 */
		public function getMediator(view:UIComponent):PopUpMediator
		{
			if (!_mediators.hasOwnProperty(view.uid)) {	
				var mediator:PopUpMediator = new PopUpMediator(view);
				_mediators[view.uid] = mediator;
			} 
			return _mediators[view.uid];
		}
	}
}

class SingletonEnforcer{}